Latvijas 35. Atklātās fizikas olimpiādes atbilžu lapa

2. uzdevums

```
2 / 3
```

3. uzdevums

```
In[1]:= eqs3 = {1 / d + 1 / f == 1 / F, f == 2 d}
```

$$\text{Out[1]}= \left\{\frac{1}{d}+\frac{1}{f}==\frac{1}{F},\ f==2\,d\right\}$$

```
In[2]:= sol3 = Solve[eqs3, {F, f}]
```

$$\text{Out[2]}= \left\{\left\{F\to\frac{2\,d}{3},\ f\to 2\,d\right\}\right\}$$

```
In[3]:= sol3 /. d → 30 cm
Out[3]= {{F → 20 cm, f → 60 cm}}
```

4. uzdevums

```
In[5]:= eq4 = u i * (1 - f) == 2 Plamp;
In[8]:= sol4 = Solve[eq4, u]
```

$$\text{Out[8]}= \left\{\left\{u\to-\frac{2\,\text{Plamp}}{(-1+f)\,i}\right\}\right\}$$

```
In[9]:= sol4 /. f → 0.1 /. i → 2 /. Plamp → 40
Out[9]= {{u → 44.4444}}
```

5.uzdevums

```
In[10]:= eqs5 = {v1 (t + t1) == v3 t1, v2 (t + t1 + t0) == v3 (t1 + t0)}
Out[10]= {(t + t1) v1 == t1 v3, (t + t0 + t1) v2 == (t0 + t1) v3}
In[11]:= Solve[Last[Last[Simplify[eqs5 /. Solve[eqs5[[1]], t1]]]], v3]
```

$$\text{Out[11]}= \left\{\left\{v3\to\frac{-t\,v1+t0\,v1+t\,v2+t0\,v2-\sqrt{-4\,t0^2\,v1\,v2+(t\,v1-t0\,v1-t\,v2-t0\,v2)^2}}{2\,t0}\right\},\right.$$
$$\left.\left\{v3\to\frac{-t\,v1+t0\,v1+t\,v2+t0\,v2+\sqrt{-4\,t0^2\,v1\,v2+(t\,v1-t0\,v1-t\,v2-t0\,v2)^2}}{2\,t0}\right\}\right\}$$

```
In[12]:= Eliminate[eqs5, t1]
```

$$\text{Out[12]}= t0\left(-v1\,v2+v1\,v3+v2\,v3-v3^2\right)==t\,(v1-v2)\,v3$$

In[15]:=
```
sol5 = Simplify[Solve[Eliminate[eqs5, t1], v3]]
```

Out[15]=
$$\left\{\left\{v3 \to \frac{-t\,v1 + t0\,v1 + t\,v2 + t0\,v2 - \sqrt{-4\,t0^2\,v1\,v2 + (t\,(-v1 + v2) + t0\,(v1 + v2))^2}}{2\,t0}\right\},\right.$$
$$\left.\left\{v3 \to \frac{-t\,v1 + t0\,v1 + t\,v2 + t0\,v2 + \sqrt{-4\,t0^2\,v1\,v2 + (t\,(-v1 + v2) + t0\,(v1 + v2))^2}}{2\,t0}\right\}\right\}$$

In[13]:=
```
num5 = {v1 → 7, v2 → 10, t → 1 / 2, t0 → 3 / 2}
```

Out[13]=
$$\left\{v1 \to 7,\ v2 \to 10,\ t \to \frac{1}{2},\ t0 \to \frac{3}{2}\right\}$$

In[16]:=
```
sol5 /. num5
```

Out[16]=
$$\left\{\left\{v3 \to \frac{1}{3}\left(27 - 3\sqrt{11}\right)\right\},\ \left\{v3 \to \frac{1}{3}\left(27 + 3\sqrt{11}\right)\right\}\right\}$$

In[17]:=
```
N[%]
```

Out[17]=
```
{{v3 → 5.68338}, {v3 → 12.3166}}
```

In[20]:=
```
Simplify[v3 > v1 && v3 > v2 /. sol5 /. num5]
```

Out[20]=
```
{False, True}
```

■ 6. uzdevums

In[21]:=
```
{(0 + 1 / 2) / 2 , (1 / 2 + 1) / 2 }^2 H
```

Out[21]=
$$\left\{\frac{H}{16},\ \frac{9\,H}{16}\right\}$$

In[22]:=
```
% /. H → 1.
```

Out[22]=
```
{0.0625, 0.5625}
```

■ 7. uzdevums

In[23]:=
```
eqs7 = {(h / v1) F == m (v1 - v2), F > M g}
```

Out[23]=
$$\left\{\frac{F\,h}{v1} == m\,(v1 - v2),\ F > g\,M\right\}$$

In[24]:=
```
num7 = {h → 0.1, v1 → 100, v2 → 99.95, m → 0.01, g → 10, M → 0.1}
```

Out[24]=
```
{h → 0.1, v1 → 100, v2 → 99.95, m → 0.01, g → 10, M → 0.1}
```

In[25]:=
```
Solve[eqs7[[1]], F]
```

Out[25]=
$$\left\{\left\{F \to \frac{m\,v1\,(v1 - v2)}{h}\right\}\right\}$$

In[26]:=
```
sol7 = Solve[eqs7[[1]], F] /. num7
```

Out[26]=
```
{{F → 0.5}}
```

In[27]:= **M g /. num7**

Out[27]= 1.

In[28]:= **eqs7[[2]] /. sol7 /. num7**

Out[28]= {False}

■ 8. uzdevums

In[29]:= **eqs8 = {Q + c t m1 ΔT == P1 t, Q + c t m2 ΔT == P2 t}**

Out[29]= $\{Q + c\ m1\ t\ \Delta T == P1\ t,\ Q + c\ m2\ t\ \Delta T == P2\ t\}$

In[30]:= **sol8 = Solve[eqs8[[2]] /. Solve[eqs8[[1]], t], ΔT]**

Out[30]= $\left\{\left\{\Delta T \to \frac{P1 - P2}{c\ (m1 - m2)}\right\}\right\}$

In[31]:= **crule = {c → (7 / 2) R / μ}**

Out[31]= $\left\{c \to \frac{7\ R}{2\ \mu}\right\}$

In[32]:= **num8 = {P1 → 10^3, P2 → 2 × 10^3, m1 → 0.15, m2 → 0.2, R → 8.31, μ → 29 * 10^(-3), t1 → 20}**

Out[32]= $\left\{P1 \to 1000,\ P2 \to 2000,\ m1 \to 0.15,\ m2 \to 0.2,\ R \to 8.31,\ \mu \to \frac{29}{1000},\ t1 \to 20\right\}$

In[33]:= **sol8 /. crule /. num8**

Out[33]= {{ΔT → 19.9416}}

In[34]:= **(t1 + ΔT) /. sol8 /. crule /. num8**

Out[34]= {39.9416}

■ 9. uzdevums

In[35]:= **Welastic[x_] := k (x - x0)^2 / 2;**
Welectrostatic[x_] := Q^2 / x / (4 π ϵ0)

In[38]:= **eqs9 = Welastic[L] + Welectrostatic[L] == Welastic[4 L] + Welectrostatic[4 L]**

Out[38]= $\frac{1}{2}\ k\ (L - x0)^2 + \frac{Q^2}{4\ L\ \pi\ \epsilon 0} == \frac{1}{2}\ k\ (4\ L - x0)^2 + \frac{Q^2}{16\ L\ \pi\ \epsilon 0}$

In[39]:= **solk = ((solkx0 = Solve[eqs9, k]) /. x0 → 2 L)**

Out[39]= $\left\{\left\{k \to \frac{Q^2}{8\ L^3\ \pi\ \epsilon 0}\right\}\right\}$

In[40]:= **toshow =**
({Welastic[x], Welectrostatic[x], Welastic[x] + Welectrostatic[x]} /. First[solk] /. Q → 1 /.
ϵ0 → (1 / 4 / π) /. x0 → 2 L /. L → 1)

Out[40]= $\left\{\frac{1}{4}\ (-2 + x)^2,\ \frac{1}{x},\ \frac{1}{4}\ (-2 + x)^2 + \frac{1}{x}\right\}$

In[41]:=
```
xeq = First[x /. FullSimplify[
    Solve[(D[Welastic[x] + Welectrostatic[x] /. solkx0, x] == 0) /. x0 → 2 L, x], L > 0]]
```

Out[41]=
$$L\,\text{Root}\left[-2-2\,\#1^2+\#1^3\,\&,\,1\right]$$

In[42]:=
```
N[%]
```

Out[42]=
$$2.3593\,L$$

In[43]:=
```
Plot[toshow, {x, 0.4, 5}, GridLines → {{1, 2, 4, xeq /. L → 1}, {Evaluate[toshow[[3]] /. x → 1]}},
 Frame → True, FrameLabel → {"x/L", "W(x)"}, PlotStyle → {Thin, Thin, Thick}]
```

Out[43]=